
Mobile BASIC is a programming language for mobile phones and PDA supporting the MIDP-1.0 / CLDC-1.0 environment. This manual documents the commands, operators and functions that are available to a Mobile BASIC programmer.

Operators

Operators are used during expression evaluation to indicate the computations that are performed on variables and constants. There are four main groups of operators that are used within Mobile BASIC and these are: Arithmetic Operators, Relational Operators, Logical Operators and Bitwise Operators. In addition there are also parenthesis operators that allow you to change the order of evaluation.

Operators are either binary or unary operators. Binary operators work on two data items, the expression, variable or constant immediately before the operator, together with the expression, variable or constant that follows the operator. Unary operators work on the expression, variable or constant that follows the unary operator.

Arithmetic Operators

Addition Operator

FORMAT:
EXP + EXP

EXAMPLE
PRINT 14 + 2

Subtraction Operator

FORMAT:
EXP - EXP

EXAMPLE
PRINT 14 - 2

Multiplication Operator

FORMAT:
EXP * EXP

EXAMPLE
PRINT 14 * 2

Division Operator

FORMAT:
EXP * EXP

EXAMPLE
PRINT 14 / 2

Power Operator

FORMAT:
EXP ^ EXP

EXAMPLE
PRINT 14 ^ 2

Unary Minus

FORMAT:
- EXP

EXAMPLE
PRINT – ( 2 * 2 )

The unary minus operator simply changes the sign of the expression, variable or constant that follows. In the given example, 2*2 will produce 4. The unary minus changes the sign from positive to negative resulting in "-4" being printed. Similarly, if the unary minus operator was applied to a negative value then the result would be a positive value.

Relational Operators

Equality Operator

FORMAT:
EXP = EXP

EXAMPLE
IF (14+2)=16 THEN PRINT "SIXTEEN"

Inequality Operator

FORMAT:
EXP <> EXP

EXAMPLE
IF (14+1)<>16 THEN PRINT "NOT EQUAL TO SIXTEEN"

Less than Operator

FORMAT:
EXP < EXP

EXAMPLE
IF (14+1)<16 THEN PRINT "LESS THAN SIXTEEN"

Less than or Equal Operator

FORMAT:
EXP <= EXP

EXAMPLE
IF (14+2)<=16 THEN PRINT "LESS THAN OR EQUAL TO SIXTEEN"

Greater than Operator

FORMAT:
EXP > EXP

EXAMPLE
IF (14+3)>16 THEN PRINT "GREATER THAN SIXTEEN"

Greater than or Equal Operator

FORMAT:
EXP >= EXP

EXAMPLE
IF (14+2)>=16 THEN PRINT "GREATER THAN OR EQUAL TO SIXTEEN"

Logical Operators

Logical operators are usually used to combine the results of relational operators in order to make more complex expressions. The result of a logical operator will be either true (1) or false (0).

NOT Operator

FORMAT:
NOT EXP

EXAMPLE
PRINT NOT 1

The NOT operator is a unary operator that takes operators on the result of the expression, variable or constant that follows. The NOT operator will either produce a zero result if the input was non zero, or a one result if the input was zero.

AND Operator

FORMAT:
EXP AND EXP

EXAMPLE
IF (A=1) AND (B=2) THEN PRINT "BOTH CONDITIONS ARE TRUE"

OR Operator

FORMAT:
EXP OR EXP

EXAMPLE
IF (A=1) OR (B=2) THEN PRINT "ONE OR MORE CONDITIONS ARE TRUE"

Bitwise Operators

Bitwise AND Operator

FORMAT:
EXP BITAND EXP

EXAMPLE
PRINT 255 BITAND 8: REM 11111111 BITAND 00001000 = 00001000

Bitwise OR Operator

FORMAT:
EXP BITOR EXP

EXAMPLE
PRINT 8 BITOR 4: REM 00001000 BITOR 00000100 = 00001100

Bitwise Exclusive OR Operator

FORMAT:
EXP BITOR EXP

EXAMPLE
PRINT 8 BITXOR 12: REM 00001000 BITXOR 00001100 = 00000100

Parenthesis Operators

The parenthesis operators are used to change the normal order of expression evaluation. There are two parenthesis operators: the left parenthesis '(' and the right parenthesis ')'. The number of left and right parenthesis used within an expression must always be equal.

When would you use parenthesis? Take the expression "3 * 4 + 2" as an example. This expression contains no parenthesis, it first multiplies 3 by 4, and then adds 2. If parenthesis where added to the expression it would be equivalent to "(3 * 4) + 2". Parenthesis would be used if you wanted the 4 and 2 to be added together before multiplying by 3. You would simply surround the "4 + 2" with left and right parenthesis resulting in the following expression: "3 * ( 4 + 2 )".

Operator Precedence

Precedence (Lower Precedence to Higher Precedence)

Commands

BLIT

FORMAT:
BLIT fromX, fromY, width, height, toX, toY

EXAMPLE:

Causes an rectangular area of the screen specified by (fromX, fromY, w, h) to be copied to (toX, toY). This command can be used to perform scrolling.

10 REM SCROLL SCREEN ONE PIXEL LEFT

20 W=WIDTH()

30 H=HEIGHT()

40 BLIT 0, 0, W, H, -1, 0

10 REM SCROLL SCREEN ONE PIXEL RIGHT

20 W=WIDTH()

30 H=HEIGHT()

40 BLIT 0, 0, W, H, 1, 0

10 REM SCROLL SCREEN ONE PIXEL UP

20 W=WIDTH()

30 H=HEIGHT()

40 BLIT 0, 0, W, H, 0, -1

10 REM SCROLL SCREEN ONE PIXEL DOWN

20 W=WIDTH()

30 H=HEIGHT()

40 BLIT 0, 0, W, H, 0, 1

CLOSE #

FORMAT:
CLOSE #channel

channel
Channel is a integer value between 0 and 9. A channel is associated with a specific file when a file is opened.

CLS

FORMAT:
CLS

EXAMPLE:
CLS

CLS causes the screen to be cleared. 

10 REM CLEARS SCREEN

20 CLS
DATA

FORMAT:
DATA comma separated list of numbers and text

EXAMPLE
DATA 56,Text String,57.0E+2

Note that the DATA statement must be on a line by itself.

Example of good data statement

10 DATA 1,2,3

Example of bad data statement – Not on its own line

10 A=1:DATA 1,2,3

DATA statements are commonly used to initialise array variables. As an example, how do you populate a string array called MONTH$ with the months of the year. One option is to explicitly assign the appropriate month to each array element. The example below illustrates how to populate the array using DATA statements.

1000 DIM MONTH$(12)

1010 FOR I%=0 TO 11

1020 READ M$

1030 MONTH$(I%)=M$

1040 NEXT I%

1100 DATA January,February,March,April,May,June

1110 DATA July,August,September,October,November,December

DEG

FORMAT:
DEG

EXAMPLE:
DEG

All subsequent trig function work in degrees. To switch to radians use the RAD command.

DELETE

FORMAT:
DELETE filenameString

EXAMPLE:
DELETE "test.dat"

DIM

FORMAT:
DIM {floatArray|integerArray|stringArray$}(integerDimension)

EXAMPLE:
DIM I%(5)



DIM F(6)



DIM A$(7)

Normally variables are associated with a single value, however, there are occasions when a variable needs to contain several values. The DIM statement is used to initialise the number of items it contains. Each item is accessed by appending the element number enclosed within parenthesis to the variable name. Array elements are indexed from zero up to the array dimension less one. Thus DIM I(3) defines 3 items: I(0), I(1) and I(2). Access any of subscripts will cause a ARRAY_BOUNDS error to be generated.

Take for instance a string variable called MONTH$. There are twelve months in the year so we declare MONTH$ to be an array consisting of 12 values.

1000 DIM MONTH$(12)

1010 MONTH$(0)="January"

1020 MONTH$(1)="February"

...

1120 MONTH$(12)="December"

In many cases it is more convenient to initialise arrays using READ / DATA statements.

DIR

FORMAT:
DIR

EXAMPLE:
DIR

This command simply produces the list of available files.

DRAWARC

FORMAT:
DRAWARC x, y, w, h, startAngle, arcAngle

EXAMPLE:

DRAWGEL

FORMAT:
DRAWGEL gelname, x, y

EXAMPLE:
DRAWGEL "image", 0, 0

The DRAWGEL command is used to draw the GEL (Graphics Element) directly into the background image. This can be used to define pseudo character based backgrounds.

1000 GELLOAD "CHAR1", 0, 0, 8, 8

1010 GELLOAD "CHAR2", 8, 0, 8, 8

1020 GELLOAD "CHAR3", 16, 0, 8, 8

1030 WIDTH%=SCREENWIDTH(0)

1040 FOR I%=0 TO WIDTH% STEP 8

1050 DRAWGEL "CHAR1", I%, 16

1060 DRAWGEL "CHAR2", I%, 24

1070 DRAWGEL "CHAR3", I%, 32

1080 NEXT I%

DRAWLINE

FORMAT:
DRAWLINE fromX, fromY, toX, toY

EXAMPLE:

The DRAWTO command causes a line of pixels to be draw from the last pixel plotted to the pixel specified by (x, y). The line is drawn using the current colour as set by the SETCOLOR command.

10 REM DRAW DIAGONAL GREEN CROSS ON SCREEN

20 CLS

30 W=WIDTH()

40 H=HEIGHT()

50 SETCOLOR 0, 255, 0

60 DRAWLINE 0,0,W,H

70 DRAWLINE W,0,0,H

DRAWRECT

FORMAT:
DRAWRECT x, y, w, h

EXAMPLE:

DRAWROUNDRECT

FORMAT:
DRAWROUNDRECT x, y, w, h, arcWidth, arcHeight

EXAMPLE:

DRAWSTRING

FORMAT:
DRAWSTRING string, x, y

EXAMPLE:

EDIT

FORMAT:
EDIT lno

EXAMPLE:
EDIT 20

Edit> 20 PRINT "Edit Me"

END

FORMAT:
END

EXAMPLE:
END

Causes the executing program to immediately finish.

1000 PRINT 1

1010 END

1020 PRINT 2

The above program prints out the value 1. It does not print 2 because the program is terminated by the END statement before line 1020 is reached.

ENTER

FORMAT: ENTER {filename | url}

EXAMPLE:
ENTER "LocalFile.lis"



ENTER "http://wwww.mobilebasic.com/GraphicsTest.lis"

FILLARC

FORMAT:
FILLARC x, y, w, h, startAngle, arcAngle

EXAMPLE:

FILLRECT

FORMAT:
FILLRECT x, y, w, h

EXAMPLE:

FILLROUNDRECT

FORMAT:
FILLROUNDRECT x, y, w, h, arcWidth, arcHeight

EXAMPLE:

FOR / TO / STEP / NEXT

FORMAT:
FOR var%=n1 TO n2 [STEP n3] … NEXT var

EXAMPLE:
FOR I%=1 TO 10: PRINT I%:NEXT I%

var, n1, n2 and n3 must all be integer.

10 FOR X=1 TO 10

20 PRINT X

30 NEXT X

The above example counts from 1 through to 10 in steps of 1. If you don't want to count in steps other than 1 then you must specify it using "STEP n". For example, to count down from 99 through to 11 in steps of –11 you should enter:-

10 FOR X=99 TO 11 STEP -11

20 PRINT X

30 NEXT X

It is also possible to nest FOR NEXT loops within another FOR NEXT LOOP.

10 FOR X=1 TO 10

20 PRINT X

30 FOR Y=1 TO 10

40 PRINT Y

50 NEXT Y

60 NEXT X

GELLOAD

FORMAT:
GELLOAD gelName, x, y, w, h

EXAMPLE:
GELLOAD "Name", 1, 1, 10, 10

The GELLOAD command creates the named Graphics Elements (GEL) and loads it with data from the specified rectangle on the canvas. GEL Images are associated with a SPRITE using the SPRITEGEL command. The sprite may then be moved around the screen using SPRITEMOVE command.

GET #

FORMAT:
GET #channel, intVariable

channel
Channel is a integer value between 0 and 9. A channel is associated with a specific file when a file is opened.

intVariable
This is a integer variable which will be used to receive the byte value read from the file associated with the channel.

GOTO

FORMAT:
GOTO lno | variable | expression

EXAMPLE:
GOTO 50



GOTO LINE%

Causes execution to proceed with the line number specified by "lno".

10 PRINT 1

20 GOTO 40

30 PRINT 2

40 PRINT 3

The above program prints 1 and 3. It does not print 2 because line 30 never gets executed. The GOTO statement at line 20 causes control to pass directly to line 40.

You can also use a variable to specify a line number as the following example illustrates.

10 FOR I%=30 TO 50 STEP 10

20 GOTO I%

30 PRINT "Line Thirty"

35 GOTO 60

40 PRINT "Line Forty"

45 GOTO 60

50 PRINT "Line Fifty"

60 NEXT I%

GOSUB

FORMAT:
GOSUB lno | variable | expression

EXAMPLE:
GOSUB 50



GOSUB LINE%

Pushes the current program location onto the return stack and then causes execution to proceed with the line number specified by "lno". You may resume execution after the GOSUB command at any time by issuing the RETURN command. Should your program decide not to return then you must use the "POP" command to remove the return address from the stack.

IF / THEN

FORMAT:
IF relationalExpression THEN statement

EXAMPLE:
IF I%<20 THEN GOTO 50

relationalExpression may involve either integer, float or string operands. You can only compare integer with integer, float with float and string with string.

IF A=B THEN GOTO 10

IF A<>B THEN GOTO 20

IF A<B THEN GOTO 30

IF A<=B THEN GOTO 40

IF A>B THEN GOTO 50

IF A>B THEN GOTO 60

IF A$=B$ THEN GOTO 10

IF A$<>B$ THEN GOTO 20

IF A$<B$ THEN GOTO 30

IF A$<=B$ THEN GOTO 40

IF A$>B$ THEN GOTO 50

IF A$>B$ THEN GOTO 60

INPUT

FORMAT: 
INPUT promptString,{floatVariable | integerVariable | stringVariable}

10 INPUT "Float: ",F

20 INPUT "Integer: ",I%

30 INPUT "String: ",S$

INPUT #

FORMAT:
INPUT #channel, intVariable



INPUT #channel, floatVariable



INPUT #channel, stringVariable

channel
Channel is a integer value between 0 and 9. A channel is associated with a specific file when a file is opened.

intVariable
This is a integer variable which will be used to receive the value read from the file associated with the channel.

floatVariable
This is a float variable which will be used to receive the value read from the file associated with the channel.

stringVariable
This is a string variable which will be used to receive the value read from the file associated with the channel.

Note: it is your program's responsibility to read byte, int, float, or string values at the correct time.

LIST

FORMAT: 
LIST { filename | [lno, [lno2]] }

EXAMPLE:
LIST "LocalFile.lis"



LIST "http://www.mobilebasic.com/upload.exe?filename=xyz"



LIST 100, 200



LIST 20



LIST

LOAD

FORMAT: 
LOAD {filename | url}

EXAMPLE:
LOAD "LocalFile.bas"



LOAD "http://wwww.mobilebasic.com/GraphicsTest.bas"

Loads programs that were saved in the tokenised format. Tokenised programs occupy less space and load much quicker since they don't need to be interpreted again. Tokenised programs are saved using the SAVE command.

NEW

FORMAT:
NEW

EXAMPLE:
NEW

Delete program from memory.

NOTE #

FORMAT:
NOTE #channel, offset

channel
Channel is a integer value between 0 and 9. A channel is associated with a specific file when a file is opened.

offset
This is a integer variable where the NOTE command stores the current offset in the file. An offset of 0 indicates the start of the file.

. File offsets are zero based – i.e. The first byte in a file is located at offset 0. The POINT command is used to move to a specific byte in a file.

NOTE is used to obtain the current file offset. The returned value is the offset where the next read or write will occur. The NOTE and POINT commands are used together to perform Random Access File I/O.

Important Note: Earlier versions of Mobile BASIC (1.4 and earlier) were record orientated. The values returned by NOTE and POINT indicated the record number. You could only select which record to read/write  - not the specific byte within the recird.

OPEN #

FORMAT:
OPEN #channel,filename,mode

EXAMPLE:
OPEN #1,"InputFile","INPUT"


OPEN #2,"OutputFile","OUPUT"

channel
Channel is a integer value between 0 and 9. Further access to this file is performed by specifying the channel.

filename
This is a string value containing the name of the file to be opened. Local filenames may contain up to 32 characters. Filenames beginning with a "." character are hidden files You can also specify a remote filename that connect to a Internet Service, for example a Java Servlet or other CGI programs.

mode
Mode must be either INPUT or OUTPUT depending on if the file is being used to input or output data. 

The following example illustrates how to access a Internet based service. The source code for the HelloServer servlet can be found in the appendices.

1000 OPEN #1,"http://www.mobilebasic.com/servlet/HelloServer","OUTPUT"

1010 INPUT "Name: ",N$

1020 PRINT #1,N$

1030 INPUT #1,A$

1040 PRINT A$

1050 CLOSE #1

PLOT

FORMAT:
PLOT x, y

The PLOT command causes a single pixel specified by (x, y) to be plotted using the current colour. The current colour is changes using the SETCOLOR command.

10 REM PLOTS A RED PIXEL IN TOP LEFT CORNER

20 CLS

30 SETCOLOR 255,0,0

40 PLOT 0, 0

POINT #

FORMAT:
POINT #channel, expression

channel
Channel is a integer value between 0 and 9. A channel is associated with a specific file when a file is opened.

recordNumber
This is a integer value that is used to select the next record to be read or written. The first record is always record 1. Additional records are added sequentially.

POINT is used to set the next record that will be read from or written to. This can be used in conjunction with NOTE to perform Random Access File I/O.

POINT is used to set the current file offset where the read and write operations are to be performed. The POINT and NOTE commands are used together to perform Random Access File I/O. The POINT command may be given the following values:-

< 0

Position read / write pointer to end of file. Usually used to append data to the file.

> fileSize
Position read / write pointer to end of file. Usually used to append data to the file.

0

Position read / write pointer to start of file.

n

Position read / write pointer to arbitrary position in file

Important Note: Earlier versions of Mobile BASIC (1.4 and earlier) were record orientated. The values returned by NOTE and POINT indicated the record number. You could only select which record to read/write  - not the specific byte within the recird.

POP

FORMAT:
POP

EXAMPLE:
POP

POPs the next return address from the stack. Any FOR LOOPS initiated since the GOSUB associated with the popped return address are also terminated.

PRINT

FORMAT:
PRINT expression

EXAMPLE:
PRINT 3



PRINT 4.5



PRINT "String"



PRINT intVariable%



PRINT floatVariable



PRINT stringVariable$

The PRINT command prints the result of the expression onto the display. Unlike other versions of BASIC you cannot supply a comma-separated list of items to be printed. If you need to print several items on the same line then you should use the available string functions. For example: -

1000 A%=123

1020 PRINT "A%=" + STR$(A%)

To print data to a file see the "PRINT #" command that follows.

PRINT #

FORMAT:
PRINT #channel, intValue



PRINT #channel, floatValue



PRINT #channel, stringValue

PUT #

FORMAT:
PUT #channel, byteValue

RAD

FORMAT:
RAD

EXAMPLE:
RAD

All subsequent trig function work in radians. To switch to degrees use the DEG command.

READ

FORMAT:
READ floatVariable | integerVariable | stringVariable

EXAMPLE:
READ floatVariable



READ intVariable%



READ stringVariable$

The READ command reads the next data item specified in the current DATA statement. The DATA statements are read starting from the first and going sequentially through to the last. When there is no data remaining a "Out of Data" error will occur which can be trapped using the TRAP command.

You can tell the interpreter to start reading the data from another line using the RESTORE command.

REM

FORMAT:
REM test string to end of line

EXAMPLE:
REM Date Written: 1st January 2003

Used to place comments and notes within the source code. Text following the "REM" statement is ignored by the interpreter.

RESTORE

FORMAT:
RESTORE intLineNumberExpression

EXAMPLE:
RESTORE 0



RESTORE LINE%

Tells the interpreter to start read data from the specified line onwards in response to READ commands.

RETURN

FORMAT:
RETURN

EXAMPLE:
RETURN

The return statement is used following a GOSUB command. The GOSUB command is used to temporarily start executing statements from the specified line number. The return command is used at the end of these lines to resume execution immediately after the GOSUB command.

Should your subroutine decide that it doesn't need to return then it must use the POP command to remove the return address from the stack. Failure to do so will use additional memory and since the stack only has a finite size you will eventually get a STACK OVERFLOW error.

SAVE

FORMAT: 
SAVE stringFilenameExpression

EXAMPLE:
SAVE "LocalFile.lis"



SAVE "http://www.mobilebasic.com/upload.exe?filename=xyz"

Saves program in tokenised format. Tokenised programs occupy less space and load much quicker since they don't need to be interpreted again. Tokenised programs are loaded using the LOAD command.

SETCOLOR

FORMAT:
SETCOLOR intRedExpression, intGreenExpression, intBlueExpression

Causes the specified RGB colour to be used until the next SETCOLOR command is issued. red, green and blue are all integer operands that should be between 0 and 255 inclusive. You should probably check to see if the environment has a color screen by checking the result from the ISCOLOR() function. If the environment doesn't support colour then GREY values should be specified using equal values for the red, green and blue components.

SLEEP

FORMAT:
SLEEP lno | variable | expression

EXAMPLE:
SLEEP 50



SLEEP MS%

The SLEEP command is used to introduce small delays in your program. The argument is an integer which specifies the number of milliseconds the program should wait for. If the argument is zero, or negative, then the SLEEP command causes the program to briefly yield the processor to allow other tasks to run. If there are no other tasks waiting then the BASIC program resumes immediately. 

Tip: Always use "SLEEP 0" or "SLEEP 1" in the main loop of a game – it will give the screen chance to update. Wherever possible use "SLEEP 1" in the loop since it can have a dramatic reduction on the amount of CPU processing power used – which could improve battery life.

Example 1 – No Sleep

10 IF FIRE(0)<>0 THEN GOTO 30

20 GOTO 10

30 END

The effect of not using the "SLEEP" command at all is that the program will repeatedly check for the fire key being pressed. This is may be done millions of times per seconds (depending on the phone).  There are several effects:- 1) There is less processor time to other important tasks such as updating the screen. 2) The battery life may be reduced because the CPU is being used all the time.

Example 2 – SLEEP 0
10 IF FIRE(0)<>0 THEN GOTO 40

20 SLEEP 0

30 GOTO 10

40 END

The effect of using "SLEEP 0" is similar to Example 1. The main difference is that other tasks, such as screen updates, are given chance to run when the SLEEP is executed. If there are no other tasks waiting then the BASIC program resumes immediately. As in case 1, the battery life may be reduced because the CPU is being used all the time.

Example 3 – SLEEP 1

10 IF FIRE(0)<>0 THEN GOTO 40

20 SLEEP 1

30 GOTO 10

40 END

Using "SLEEP 1" is much more friendly to the phone. It allows other tasks to run more readily and greatly reduces the amount of CPU processing power consumed. For example, lets say that your phone processor runs at 10MHz – 10 million instructions a second. It will only take the processor a small number of instructions to work out if the FIRE button has pressed - lets say its 100 instructions for sake of an example. This would result in the loop being executed 100000 times a second. Is this necessary? Do you need to detect the FIRE button being pressed within 1/100000 second? Probably not! "SLEEP 1" puts the program to SLEEP for 1/1000 second before it checks again. The result is that the program will check the FIRE button 1000 times a second using 100 instructions each time. The total number of instructions executed would be 100*1000 = 100000 which is 1% of the available processing time. The remaining 99% of the processor time is now idle which may reduce power consumption as well as giving the screen more chance to update.

SPRITEGEL

FORMAT:
SPRITEGEL spriteName, gelName

EXAMPLE:
SPRITEGEL "ShipSprite","ShipGel"

SPRITEGEL associated a named sprite with a named Graphics Element (GEL). If the sprite doesn't exist then a new sprite is created. Sprite animation can be achieved by associating the Sprite with a different GEL. The sprite can be moved around the screen using SPRITEMOVE. Collisions between sprites can be detected using the SPRITEHIT function.

SPRITEMOVE

FORMAT:
SPRITEMOVE spriteName, x, y

EXAMPLE:
SPRITEMOVE "ShipSprite", 20, 30

Causes the named sprite to be moved to the new location specified in the command.

STOP

FORMAT:
STOP

EXAMPLE:
STOP

Causes the program to immediately stop execution.

TRAP

FORMAT:
TRAP lno | variable | expression

EXAMPLE:
TRAP 9000



TRAP LINE%

Causes the specified line number to be called following an error. The error number can be obtained using the ERR() function.

Functions

All functions within Mobile BASIC are used within expressions. Each functions name is followed by a left parenthesis and a comma separated list of one or more parameters terminated by a right parenthesises.

Mathematical Functions

ABS

FORMAT:
ABS(VAL)

EXAMPLE:
X = ABS(-2.5)

Returns absolute value of the number.

ACOS

FORMAT:
ACOS(VAL)

EXAMPLE:
X = ACOS(0.5)

ASIN

FORMAT:
ASIN(VAL)

EXAMPLE:
X = ASIN(0.5)

ATAN

FORMAT:
ATAN(VAL)

EXAMPLE:
X = ATAN(0.5)

COS

FORMAT:
COS(VAL)

EXAMPLE:
X = COS(0.5)

EXP

FORMAT:
EXP(VAL)

EXAMPLE:
X = EXP(0.5)

LOG

FORMAT:
LOG(VAL)

EXAMPLE:
X = LOG(0.5)

MOD

FORMAT:
MOD(VAL1, VAL2)

EXAMPLE:
X = MOD(157,100)

SIN

FORMAT:
SIN(VAL)

EXAMPLE:
X = SIN(0.5)

SQR

FORMAT:
SQR(VAL)

EXAMPLE:
X = SQR(81.0)

TAN

FORMAT:
ATAN(VAL)

EXAMPLE:
X = ATAN(0.5)

User Interface Functions

CHOICEFORM

FORMAT:
I% = CHOICEFORM(FormTitle$, ProceedButtonText$, CancelButtonText$,




ChoiceLabel$, ChoiceArray$, Mode%)

Mode% can be 0 or 1.

Mode 0 – Exclusive Mode. Only one item can be selected. The function returns the array element selected starting from element 0. If the user terminated the screen with Cancel then the function returns –1.

Mode 1 – Multiple Selection Mode. The user may select 0, 1 or many items. The function returns an integer whose bits indicate if the item was selected. There may be a maximum of 32 items in this mode. Bit 0 = Item 1 selected, Bit 1 = Item 2 selected etc. Again the function returns –1 if the user pressed cancel. Care must be taken if there are 32 items since –1 would also be returned if all items have been selected.

DATEFORM

FORMAT:
I% = DATEFORM(FormTitle$, ProceedButtonText$, CancelButtonText$,




DateFieldLabel$, dateArray%, mode%)

mode% = 0 – Date and Time

mode% = 1 – Date Only

mode% = 2 – Time Only

dateArray% should be dimensioned to 2.

dateArray% contains input date on entry and output date on completion.

I% = 1 if user pressed proceed button, I%=0 if user pressed cancel button.

10 DIM DT%(2)

20 D%=DAYS()

30 T%=MILLISECONDS()

40 DT%(0)=D%

50 DT%(1)=T%

60 I%=DATEFORM("Date Form","Continue","Cancel","Today",DT%,0)

70 IF I%=0 THEN 100

80 D%=DT%(0)

90 T%=DT%(1)

100 END

EDITFORM

FORMAT:
A$ = EDITFORM(FormTitle$, ProceedButtonText$, CancelButtonText$,




TextFieldLabel$, DefaultText$, MaxLen%, Mode%)

Mode% can be 0 to 5.

Mode 0 – Any Text

Mode 1 – Password

Mode 2 – Numeric

Mode 3 – Email Addresses

Mode 4 – Phone Numbers

Mode 5 - URL

GAUGEFORM

FORMAT:
I% = GAUGEFORM(FormTitle$, ProceedButtonText$, CancelButtonText$,




GaugeLabel$, MaxValue%, InitialValue%, mode%)

mode% may be 0 or 1. Mode 0 is a non interactive gauge. Mode 1 is an interactive gauge. MaxValue% must be an integer number greater than 1.

MESSAGEFORM

FORMAT:
A$ = MESSAGEFORM(FormTitle$, ProceedButtonText$, CancelButtonText$,




MessageLabel$, Message$)

Graphics Functions

ISCOLOR

FORMAT:
ISCOLOR(DUMMY%)

EXAMPLE:
X = ISCOLOR (0)

Returns true (1) if screen is color otherwise it returns false(0).

NUMCOLORS

FORMAT:
NUMCOLORS(DUMMY%)

EXAMPLE:
X = NUMCOLORS (0)

Returns number of colours (or grey levels) available on device.

SCREENHEIGHT

FORMAT:
SCREENHEIGHT(DUMMY%)

EXAMPLE:
HEIGHT = SCREENHEIGHT(0)

Returns screen height in pixels.

SCREENWIDTH

FORMAT:
SCREENWIDTH(DUMMY%)

EXAMPLE:
WIDTH = SCREENWIDTH(0)

Returns screen width in pixels.

SPRITEHIT

FORMAT:
SPRITEHIT (spriteName1, spriteName2)

EXAMPLE:
HITFLAG% = SPRITEHIT("Ship", "Missile")

The SPRITEHIT function returns 1 if the two sprites overlap or 0 if the two sprites do not overlap. This function doesn't test for collision on a pixel basis.

STRINGHEIGHT

FORMAT:
STRINGHEIGHT(string)

EXAMPLE:
HEIGHT = STRINGHEIGHT("Message")

STRINGWIDTH

FORMAT:
STRINGWIDTH(string)

EXAMPLE:
WIDTH = STRINGWIDTH("Message")

Data and Time Functions

YEAR

FORMAT:
YEAR(DAY%,MS%)

EXAMPLE:
X = YEAR(DAY%,MS%)

Returns year associated with DATE% and MS% as an Integer.

MONTH

FORMAT:
MONTH(DAY%,MS%)

EXAMPLE:
X = MONTH(DAY%,MS%)

Returns month associated with DAY% and MS% as an Integer. Month values range from 1 to 12 and represent the months January to December.

DAY

FORMAT:
DAY(DAY%,MS%)

EXAMPLE:
X = DAY(DAY%,MS%)

Returns day of month associated with DAY% and MS% as an Integer. Day values range from 1 to 31 depending on the month and year.

DAYS

FORMAT:
DAYS(DUMMY%)

EXAMPLE:
X = DAY(0)

Returns the number of days since the epoch as an Integer.

HOUR

FORMAT:
HOUR(DAY%,MS%)

EXAMPLE:
X = HOUR(DAY%,MS%)

Returns hour within the day associated with DAY% and MS% as an Integer between 0 and 23.

MINUTE

FORMAT:
MINUTE(DAY%,MS%)

EXAMPLE:
X = MINUTE(DAY%,MS%)

Returns minute within the hour associated with DAY% and MS% as an Integer between 0 and 59.

SECOND

FORMAT:
SECOND(DAY%,MS%)

EXAMPLE:
X = SECOND(DAY%,MS%)

Returns second within the  minute associated with DAY% and MS% as an Integer between 0 and 59.

MILLISECOND

FORMAT:
MILLISECOND(DAY%,MS%)

EXAMPLE:
X = MILLISECOND(DAY%,MS%)

Returns millisecond within the second associated with DAY% and MS% as an Integer between 0 and 999.

MILLISECONDS

FORMAT:
MILLISECONDS(DUMMY%)

EXAMPLE:
X = MILLISECONDS(0)

Returns number of milliseconds elapsed since midnight as an Integer.

Game Functions

The MIDP specification defines various facilities that are useful for games programming. These take the form of Up, Down, Left, Right and Fire buttons together with 4 game specific keys known as GAME A, GAME B, GAME C and GAME D. How these features are physically implemented is left to the hardware designers:- For instance, UP, DOWN, LEFT, RIGHT and FIRE could be implemented using a cursor pad, a joystick or simply the phones numeric keypad.

Mobile BASIC provides access to these features via a series of functions that simply return a non zero value if the key is pressed or zero if the key is not pressed.

DOWN

FORMAT:
DOWN(DUMMY%)

EXAMPLE:
X = DOWN(0)

Returns non zero if DOWN is selected or zero if the DOWN is not selected.

FIRE

FORMAT:
FIRE(DUMMY%)

EXAMPLE:
X = FIRE(0)

Returns non zero if FIRE is selected or 0 if FIRE is not selected.

10 PRINT "Press Fire to Continue"

20 A=FIRE()

30 IF A=0 THEN 20

40 STOP

GAMEA

FORMAT:
GAMEA(DUMMY%)

EXAMPLE:
X = GAMEA(0)

Returns non zero if GAMEA is selected or 0 if GAMEA is not selected.

GAMEB

FORMAT:
GAMEB(DUMMY%)

EXAMPLE:
X = GAMEB(0)

Returns non zero if GAMEB is selected or 0 if GAMEB is not selected.

GAMEC

FORMAT:
GAMEC(DUMMY%)

EXAMPLE:
X = GAMEC(0)

Returns non zero if GAMEC is selected or 0 if GAMEC is not selected.

GAMED

FORMAT:
GAMED(DUMMY%)

EXAMPLE:
X = GAMED(0)

Returns non zero if GAMED is selected or 0 if GAMED is not selected.

LEFT

FORMAT:
LEFT(DUMMY%)

EXAMPLE:
X = LEFT(0)

Returns non zero if LEFT is selected or 0 if LEFT is not selected.

RIGHT

FORMAT:
RIGHT(DUMMY%)

EXAMPLE:
X = RIGHT(0)

Returns non zero if RIGHT is selected or 0 if RIGHT is not selected.

UP

FORMAT:
UP(DUMMY%)

EXAMPLE:
X = UP(0)

Returns non zero if UP is selected or 0 if UP is not selected.

String Functions

String variables are denoted by the variable name ending with a "$" character. String constants are defined by a text string enclosed within double quotation marks.

ASC

FORMAT:
ASC(STRING)

EXAMPLE:
X% = ASC("A"): REM X%=65

CHR$

FORMAT:
CHR$(CHARCODE)

EXAMPLE:
X$ = CHR$(65): REM X$="A"

LEFT$

FORMAT:
LEFT$(STRING, NCHARS)

EXAMPLE:
X$ = LEFT$("1234567890", 5): REM X$="12345"

LEN

FORMAT:
LEN(STRING)

EXAMPLE:
X% = LEN("1234567890"): REM X%=10

MID$

FORMAT:
MID$(STRING,START,NCHARS)

EXAMPLE:
X$ = MID$("1234567890", 3,6): REM X$="345678"

RIGHT$

FORMAT:
RIGHT$(STRING, NCHARS)

EXAMPLE:
X$ = RIGHT$("1234567890", 5): REM X$="67890"

STR$

FORMAT:
STR$(NUM)

EXAMPLE:
X$ = STR$(123): REM X$="123"



X$ = STR$(1.234): REM X$="1.234"

VAL

FORMAT:
VAL(STRING)

EXAMPLE:
X% = VAL("123"): REM X%=123



X = VAL("1.23"): REM X = 1.23

Miscellaneous Functions

ERR

FORMAT:
ERR(DUMMY%)

EXAMPLE:
X% = ERR(0)

ERR() is used within an error handling routine to identify the last error number that occurred.

FRE

FORMAT:
FRE(arg%)

EXAMPLE:
X% = FRE(0)

FRE() returns various memory statistics depending on the value of arg% :-

0. Byte remaining for storage of BASIC program

1. Size of current BASIC program in bytes

2. Maximum amout of bytes available for storage of BASIC program

3. Amount of free memory, in bytes, available on the phone

4. Total memory in bytes, available on the phone

PROPERTY$

FORMAT:
PROPERTY$("propertyName")

EXAMPLE:
FILENAME% = READDIR$("microedition.platform")

PROPERTY$() returns the string value associated with the specified property. If the phone doesn't support the property then the function returns the empty string "".

Common Property Values are: -

1. microedition.platform

2. microedition.configuration

3. microedition.profiles

4. microedition.locale

5. microedition.encodingClass

6. microedition.http_proxy

7. microedition.encoding

10 P$=PROPERTY$("microedition.platform")

20 PRINT P$

The program prints the following on a Nokia 3510i: - "Nokia3510I"

Note: Not all phones support all of the above properties and some phone may support additional properties not listed above.

READDIR$

FORMAT:
READDIR$(FILTER$)

EXAMPLE:
FILENAME% = READDIR$("*")

READDIR$() is used to return the files available to Mobile BASIC. The directory is read each time READDIR$() is called with a non blank filter. READDIR$() return either a filename or a blank string if all files have been returned. The following example illustrates how to print out the available filenames. Filenames starting with "." are hidden files which are not returned by READDIR$().

10 F$=READDIR$("*")

20 IF F$<>"" THEN PRINT F$:F$=READDIR$(""):GOTO 20

RND

FORMAT:
RND(DUMMY%)

EXAMPLE:
X% = RND(0)

Returns 32 bit Random Number. If you need smaller random numbers then you can use the MOD function. For example, to return random numbers between 0 and 99 use "R=RND()" followed by "R=MOD(R%,100)"

Error Numbers

General Error Numbers

1. STACK EMPTY – The RETURN, NEXT and POP statements remove items from the internal stack. This error indicates that the stack is empty and there is nothing to remove.

2. STACK OVERFLOW – GOSUB and FOR NEXT LOOPS push items onto the internal stack. This error indicates that there is not enough room on the stack to continue.

3. LINE NOT FOUND – The GOTO, GOSUB or TRAP statement has been supplied with a line number that does not exist in the program.

4. NEXT BEFORE FOR – The NEXT statement is always matched with a FOR statement. This error means that no FOR statement has been executed.

5. ARRAY BOUNDS – The program is either trying to dimension an array with invalid bounds or it is trying to access an element outside the declared range.

6. VALUE ERROR – A non numerical value has been used where a numerical value was expected. For example, enterering "ABC" in response to an INPUT statement that is expecting a number.

7. INTEGER EXPECTED

8. FLOAT EXPECTED

9. STRING EXPECTED

10. INTEGER ARRAY EXPECTED

11. FLOAT ARRAY EXPECTED

12. STRING ARRAY EXPECTED

13. LVALUE EXPECTED

14. RVALUE EXPECTED

15. DATA LINE ERROR – The DATA statement must be on a line by itself. If it isn't then an error will be generated during program execution.

16. OUT OF DATA – You have issued a READ command but there is either no DATA statements or they have all been READ. You can reposition where items are READ from using the RESTORE command.

17. INCORRECT NUMBER OF ARGUMENTS

18. PARENTHESIS NESTING ERROR

19. EXPRESSION INCOMPLETE

20. HASH EXPECTED

I/O Error Numbers

4096. INVALID CHANNEL – Channel must be 0 .. 9 inclusive.

4097. INVALID I/O MODE

4098. CHANNEL ALREADY IN USE – The I/O channel you are trying to OPEN is already in use. You should close I/O channels after they have been finished with.

4099. CHANNEL NOT OPEN – The I/O channel you are trying to read/write is not open.

4100. FILE NOT FOUND – The file could not be found.

4101. I/O ERROR – General I/O Error

Other Errors

-1
INTERNAL ERROR

